
3. Corrections
Change 1 – Corrections to altstep ‘a_IP_IMS_DL_Handler ()’

	Altstep name
	a_IP_IMS_DL_Handler()

	Reason for change
	In case the UE is sending a Response over UDP, the UE shall use local port pc, see TS 34.229-3 Figure 1. The used port server which maps to port ps in Figure 1 was not correct. 


	Summary of change
	In case that responses are sent by the SS to the UE, the client port shall be used.

	TTCN module
	IP_PTC\IP_PTC_IMS_Handler.ttcn

	MCC160 Comment
	


Before change

    altstep a_IP_IMS_DL_Handler(inout IMS_Server_Type p_ImsServer,

                              integer p_PdnIndex) runs on IP_PTC

  { // to handle requests or responses from the IMS PTC

    var IMS_DATA_REQ v_IMS_DATA_REQ;

    var IMS_DATA_RSP v_IMS_DATA_RSP;

    var bitstring v_Data;

    [] IMS_Server[p_PdnIndex].receive(cr_IMS_DATA_RSP_Any) -> value v_IMS_DATA_RSP

      {

        v_Data := encvalue(v_IMS_DATA_RSP.Response);

        fl_IP_IMS_DL_Send(p_ImsServer, v_IMS_DATA_RSP.RoutingInfo, server, v_Data);
      }

    [] IMS_Client[p_PdnIndex].receive(cr_IMS_DATA_REQ_Any) -> value v_IMS_DATA_REQ

      {

        v_Data := encvalue(v_IMS_DATA_REQ.Request);

        fl_IP_IMS_DL_Send(p_ImsServer, v_IMS_DATA_REQ.RoutingInfo, client, v_Data);

      }

  }

After change

  altstep a_IP_IMS_DL_Handler(inout IMS_Server_Type p_ImsServer,

                              integer p_PdnIndex) runs on IP_PTC

  { // to handle requests or responses from the IMS PTC

    var IMS_DATA_REQ v_IMS_DATA_REQ;

    var IMS_DATA_RSP v_IMS_DATA_RSP;

    var bitstring v_Data;

    [] IMS_Server[p_PdnIndex].receive(cr_IMS_DATA_RSP_Any) -> value v_IMS_DATA_RSP

      {

        v_Data := encvalue(v_IMS_DATA_RSP.Response);

     // See 34.229-3 Figure 1

    if ( v_IMS_DATA_RSP.RoutingInfo.Protocol == udp) {

          fl_IP_IMS_DL_Send(p_ImsServer, v_IMS_DATA_RSP.RoutingInfo, client, v_Data);

     } else {

          fl_IP_IMS_DL_Send(p_ImsServer, v_IMS_DATA_RSP.RoutingInfo, server, v_Data);

     }

      }
    [] IMS_Client[p_PdnIndex].receive(cr_IMS_DATA_REQ_Any) -> value v_IMS_DATA_REQ

      {

        v_Data := encvalue(v_IMS_DATA_REQ.Request);

        fl_IP_IMS_DL_Send(p_ImsServer, v_IMS_DATA_REQ.RoutingInfo, client, v_Data);

      }

  }

Change 2 – Corrections to altstep ‘a_IP_IMS_UL_ReceiveDataUDP ()’

	Altstep name
	a_IP_IMS_UL_ReceiveDataUDP()

	Reason for change
	Variables “v_NW_Address” and “v_UE_Address” were undefined.

	Summary of change
	Variables are set in all cases.

	TTCN module
	IP_PTC\IP_PTC_IMS_Handler.ttcn

	MCC160 Comment
	


Before change

  altstep a_IP_IMS_UL_ReceiveDataUDP(IMS_Server_Type p_ImsServer,

                                     integer p_PdnIndex,

                                     template (present) IP_AddrInfo_Type p_IpAddr,

                                     template (present) PortNumber_Type  p_PortNumber,

                                     IMS_ProtectedUnprotected_Type p_ProtectedUnprotected) runs on IP_PTC

  { /* receive data from the UE */

    var IP_SOCKET_IND v_IP_SOCKET_IND;

    var Datagram_Content_Type v_Data;

    var template (omit) IP_AddrInfo_Type v_NW_Address := omit;

    var template (omit) IP_AddrInfo_Type v_UE_Address := omit;

    var template (value) IMS_RoutingInfo_Type v_RoutingInfo;

    var IMS_ClientServer_Type v_ClientServer;

    // receive UDP data

    [] IP_SOCK.receive(cr_UDP_DATA_IND(cr_IP_Socket(p_IpAddr, p_PortNumber)))

      -> value v_IP_SOCKET_IND

      {

        v_Data := v_IP_SOCKET_IND.DATA.Ind.UDP.RecvFrom.Buffer;

        if (not fl_ImsServer_UeIsRegistered(p_ImsServer)) {

          /* initial access:  => IMS PTC gets handed over the UE's and the NW's IP address (=> IPv4 or IPv6) */

          v_NW_Address := v_IP_SOCKET_IND.DATA.ConnectionId.Local.IpAddr;    // may be IPv4 or IPv6

          v_UE_Address := v_IP_SOCKET_IND.DATA.ConnectionId.Remote.IpAddr;   // may be IPv4 or IPv6
          v_ClientServer := server;   // intial access is only applicable for requests from the UE

        }

        else {

          v_ClientServer := fl_IP_IMS_UL_ResponseOrRequest(v_Data);   /* for UDP Requests and Responses are sent by the UE from Port_uc to Port_ps,

                                                                       * i.e. there is not data for Port_us to Port_pc and therefore Requests and Responses

                                                                       * cannot be distinguished by the port on which they have been received */

        }

        v_RoutingInfo := cs_IMS_RoutingInfo(udp, p_ProtectedUnprotected, v_UE_Address, v_NW_Address);

        fl_IP_IMS_UL_DecodeAndSend(p_PdnIndex, v_ClientServer, v_RoutingInfo, v_Data);

      }

  }

After change

  altstep a_IP_IMS_UL_ReceiveDataUDP(IMS_Server_Type p_ImsServer,

                                     integer p_PdnIndex,

                                     template (present) IP_AddrInfo_Type p_IpAddr,

                                     template (present) PortNumber_Type  p_PortNumber,

                                     IMS_ProtectedUnprotected_Type p_ProtectedUnprotected) runs on IP_PTC

  { /* receive data from the UE */

    var IP_SOCKET_IND v_IP_SOCKET_IND;

    var Datagram_Content_Type v_Data;

    var template (omit) IP_AddrInfo_Type v_NW_Address := omit;

    var template (omit) IP_AddrInfo_Type v_UE_Address := omit;

    var template (value) IMS_RoutingInfo_Type v_RoutingInfo;

    var IMS_ClientServer_Type v_ClientServer;

    // receive UDP data

    [] IP_SOCK.receive(cr_UDP_DATA_IND(cr_IP_Socket(p_IpAddr, p_PortNumber)))

      -> value v_IP_SOCKET_IND

      {

        v_Data := v_IP_SOCKET_IND.DATA.Ind.UDP.RecvFrom.Buffer;
        v_NW_Address := v_IP_SOCKET_IND.DATA.ConnectionId.Local.IpAddr;  
        v_UE_Address := v_IP_SOCKET_IND.DATA.ConnectionId.Remote.IpAddr; 

        if (not fl_ImsServer_UeIsRegistered(p_ImsServer)) {
          /* initial access:  => IMS PTC gets handed over the UE's and the NW's IP address (=> IPv4 or IPv6) */

          v_ClientServer := server;   // intial access is only applicable for requests from the UE

        }

        else {
          v_ClientServer := fl_IP_IMS_UL_ResponseOrRequest(v_Data);   /* for UDP Requests and Responses are sent by the UE from Port_uc to Port_ps,

                                                                       * i.e. there is not data for Port_us to Port_pc and therefore Requests and Responses

                                                                       * cannot be distinguished by the port on which they have been received */

        }

        v_RoutingInfo := cs_IMS_RoutingInfo(udp, p_ProtectedUnprotected, v_UE_Address, v_NW_Address);

        fl_IP_IMS_UL_DecodeAndSend(p_PdnIndex, v_ClientServer, v_RoutingInfo, v_Data);

      }

  }

Change 3 – Corrections to function ‘fl_IP_IMS_GetPort ()’

	Function name
	fl_IP_IMS_GetPort()

	Reason for change
	The computation of the local and remote SS port numbers in case of secure ports over UDP was not complete. 


	Summary of change
	Added the missing code (see TTCN below).

	TTCN module
	IP_PTC\IP_PTC_IMS_Handler.ttcn

	MCC160 Comment
	


Before change

  function fl_IP_IMS_GetPort(IMS_Server_Type p_ImsServer,

                             IMS_ProtectedUnprotected_Type p_ProtectedUnprotected,         // protected or unprotected

                             IMS_LocalRemote_Type p_LocalRemote,                           // local or remote

                             InternetProtocol_Type p_Protocol,                             // udp or tcp

                             IMS_ClientServer_Type p_ClientServer)                         // client or server

    return PortNumber_Type

  { /* NOTE1: even when security is started it shall be possible to send e.g. the RegisterResponse on an unprotected port

     *        => we need to have p_ProtectedUnprotected and cannot just use fl_ImsServer_SecurityIsEstablished *

     * NOTE2: function returns a value even for the unprotected local tcp port when the SS is client (GIBA only);

     *        this port shall be ignored and instead an ephemeral port shall be assigned by the system */

    var IMS_ProtectedPorts_Type v_ProtectedPorts;

    var PortNumber_Type v_PortNumber;

    select (p_ProtectedUnprotected) {

      case (protected) {

        if (not fl_ImsServer_SecurityIsEstablished(p_ImsServer)) {

          FatalError(__FILE__, __LINE__, "no security established");

        }

        v_ProtectedPorts := p_ImsServer.RegistrationInfo.SecurityInfo.ProtectedPorts;

        select (p_Protocol) {

          case (udp) {

            select (p_LocalRemote) {

              case (local)  {v_PortNumber := v_ProtectedPorts.Port_ps}

              case (remote) {v_PortNumber := v_ProtectedPorts.Port_us}

            }

          }
          case (tcp) {

            select (p_LocalRemote) {

              case (local) {

                select (p_ClientServer) {

                  case (client) {v_PortNumber := v_ProtectedPorts.Port_pc}

                  case (server) {v_PortNumber := v_ProtectedPorts.Port_ps}

                }

              }

              case (remote) {

                select (p_ClientServer) {

                  case (client) {v_PortNumber := v_ProtectedPorts.Port_uc}

                  case (server) {v_PortNumber := v_ProtectedPorts.Port_us}

                }

              }

            }

          }

        }

      }

      case (unprotected) {

        select (p_LocalRemote) {

          case (local)  {v_PortNumber := p_ImsServer.UnprotectedPorts.Port_ps}

          case (remote) {v_PortNumber := p_ImsServer.UnprotectedPorts.Port_us}

        }

      }

    }

    return v_PortNumber;

  }

After change

  function fl_IP_IMS_GetPort(IMS_Server_Type p_ImsServer,

                             IMS_ProtectedUnprotected_Type p_ProtectedUnprotected,         // protected or unprotected

                             IMS_LocalRemote_Type p_LocalRemote,                           // local or remote

                             InternetProtocol_Type p_Protocol,                             // udp or tcp

                             IMS_ClientServer_Type p_ClientServer)                         // client or server

    return PortNumber_Type

  { /* NOTE1: even when security is started it shall be possible to send e.g. the RegisterResponse on an unprotected port

     *        => we need to have p_ProtectedUnprotected and cannot just use fl_ImsServer_SecurityIsEstablished *

     * NOTE2: function returns a value even for the unprotected local tcp port when the SS is client (GIBA only);

     *        this port shall be ignored and instead an ephemeral port shall be assigned by the system */

    var IMS_ProtectedPorts_Type v_ProtectedPorts;

    var PortNumber_Type v_PortNumber;

    select (p_ProtectedUnprotected) {

      case (protected) {

        if (not fl_ImsServer_SecurityIsEstablished(p_ImsServer)) {

          FatalError(__FILE__, __LINE__, "no security established");

        }

        v_ProtectedPorts := p_ImsServer.RegistrationInfo.SecurityInfo.ProtectedPorts;

        select (p_Protocol) {

          case (udp) {

            select (p_LocalRemote) {

              case (local)  {



    select (p_ClientServer) {

                  case (client) {v_PortNumber := v_ProtectedPorts.Port_pc}

                  case (server) {v_PortNumber := v_ProtectedPorts.Port_ps}

                }




  } 

              case (remote) {

                select (p_ClientServer) {

                  case (client) {v_PortNumber := v_ProtectedPorts.Port_us}

                  case (server) {v_PortNumber := v_ProtectedPorts.Port_uc}

                }


  




  }

            }

          }
          case (tcp) {

            select (p_LocalRemote) {

              case (local) {

                select (p_ClientServer) {

                  case (client) {v_PortNumber := v_ProtectedPorts.Port_pc}

                  case (server) {v_PortNumber := v_ProtectedPorts.Port_ps}

                }

              }

              case (remote) {

                select (p_ClientServer) {

                  case (client) {v_PortNumber := v_ProtectedPorts.Port_uc}

                  case (server) {v_PortNumber := v_ProtectedPorts.Port_us}

                }

              }

            }

          }

        }

      }

      case (unprotected) {

        select (p_LocalRemote) {

          case (local)  {v_PortNumber := p_ImsServer.UnprotectedPorts.Port_ps}

          case (remote) {v_PortNumber := p_ImsServer.UnprotectedPorts.Port_us}

        }

      }

    }

    return v_PortNumber;

  }
Change 4 – Corrections to function ‘fl_ImsServer_SecurityStart ()’

	Function name
	fl_ImsServer_SecurityStart()

	Reason for change
	Client port in the SS, i.e the client socket was not opened. Only the Port “v_SecurityInfo.ProtectedPorts.Port_ps” was opened, but not Port “v_SecurityInfo.ProtectedPorts.Port_pc”.


	Summary of change
	Client port in the SS is configured.

	TTCN module
	IP_PTC\IP_PTC_IMS_Handler.ttcn

	MCC160 Comment
	


Before change

…

  function fl_ImsServer_SecurityStart(IMS_RegistrationInfo_Type p_RegistrationInfo,

                                      IPsec_SecurityKeys_Type p_SecurityKeys,

                                      IP_DrbInfo_Type p_DrbInfo) runs on IP_PTC return IMS_SecurityInfo_Type

  {

    var IP_AddrInfo_Type v_LocalIpAddr := p_RegistrationInfo.NW_Address;

    var IP_AddrInfo_Type v_RemoteIpAddr := p_RegistrationInfo.UE_Address;

    var IMS_SecurityInfo_Type v_SecurityInfo := p_RegistrationInfo.SecurityInfo;

    var IMS_SPIs_Type v_SPIs := v_SecurityInfo.SPIs;

    var PortNumber_Type v_LocalPort := v_SecurityInfo.ProtectedPorts.Port_ps;

    // get SPIs for the network side:

    v_SecurityInfo.SPIs := f_SPIs_SetNetworkValues(v_SPIs);

    fl_IPsec_Configure(v_LocalIpAddr, v_RemoteIpAddr, v_SecurityInfo, p_SecurityKeys);

    f_UdpServer_Start(v_LocalIpAddr, v_LocalPort, p_DrbInfo);

    f_TcpServer_Start(v_LocalIpAddr, v_LocalPort, p_DrbInfo);

    return v_SecurityInfo;

  }

…

After change
…
function fl_ImsServer_SecurityStart(IMS_RegistrationInfo_Type p_RegistrationInfo,

                                      IPsec_SecurityKeys_Type p_SecurityKeys,

                                      IP_DrbInfo_Type p_DrbInfo) runs on IP_PTC return IMS_SecurityInfo_Type

  {

    var IP_AddrInfo_Type v_LocalIpAddr := p_RegistrationInfo.NW_Address;

    var IP_AddrInfo_Type v_RemoteIpAddr := p_RegistrationInfo.UE_Address;

    var IMS_SecurityInfo_Type v_SecurityInfo := p_RegistrationInfo.SecurityInfo;

    var IMS_SPIs_Type v_SPIs := v_SecurityInfo.SPIs;

    var PortNumber_Type v_LocalPort := v_SecurityInfo.ProtectedPorts.Port_ps;

    // get SPIs for the network side:

    v_SecurityInfo.SPIs := f_SPIs_SetNetworkValues(v_SPIs);

    fl_IPsec_Configure(v_LocalIpAddr, v_RemoteIpAddr, v_SecurityInfo, p_SecurityKeys);

    f_UdpServer_Start(v_LocalIpAddr, v_LocalPort, p_DrbInfo);

    f_UdpServer_Start(v_LocalIpAddr,v_SecurityInfo.ProtectedPorts.Port_pc, p_DrbInfo); 

    f_TcpServer_Start(v_LocalIpAddr, v_LocalPort, p_DrbInfo);

    return v_SecurityInfo;

  }   

…
Change 5 – Corrections to function ‘f_IMS_PTC_GetPort_ps ()’

	Function name
	f_IMS_PTC_GetPort_ps()

	Reason for change
	This function is used to get the local P-CSCF server port’s IP address. However, the returned value that of the UE side (“v_Protected.Port_us”).


	Summary of change
	Returned port number corrected to “v_Protected.Port_ps” (i.e. P-CSCF server port).

	TTCN module
	CommonIMS\IMS_CommonProcedures_Registration.ttcn

	MCC160 Comment
	


Before change

…

  function f_IMS_PTC_GetPort_ps(boolean p_IsProtected := true) runs on IMS_PTC return integer

  {

    var IMS_ProtectedPorts_Type v_Protected;

    var integer v_Port_ps := tsc_IMS_PortNumber_5060;

    if (p_IsProtected) {

      v_Protected := f_IMS_PTC_Security_GetProtectedPorts();

      v_Port_ps := v_Protected.Port_us; 
    }

    return v_Port_ps;

  }
…
After change

…

  function f_IMS_PTC_GetPort_ps(boolean p_IsProtected := true) runs on IMS_PTC return integer

  {

    var IMS_ProtectedPorts_Type v_Protected;

    var integer v_Port_ps := tsc_IMS_PortNumber_5060;

    if (p_IsProtected) {

      v_Protected := f_IMS_PTC_Security_GetProtectedPorts();

      v_Port_ps := v_Protected.Port_ps; 
    }

    return v_Port_ps;

  }
…
Change 6 – Corrections to function ‘f_IMS_NotifyRequest_MessageHeader_A16 ()’

	Function name
	f_IMS_NotifyRequest_MessageHeader_A16()

	Reason for change
	According to the default contents of the NOTIFY message sent by the UE, see TS 34.229-1 Annex A.1.6, the Via header shall contain for condition A1 (i.e. with IMS security) the local P-CSCF server port’s IP address (“v_Port_ps”). However, the UE’s server port IP address was used (“v_Port_us”). 


	Summary of change
	Via header corrected to use the local P-CSCF server port’s IP address of the SS side.

	TTCN module
	CommonIMS\IMS_CommonProcedures_Registration.ttcn

	MCC160 Comment
	


Before change

…

  function f_IMS_NotifyRequest_MessageHeader_A16(SUBSCRIBE_Request p_SubscribeRequest,

                                                 boolean p_CondA2,

                                                 integer p_CseqValue,

                                                 template (value) SubscriptionState p_SubscriptionState) runs on IMS_PTC return template (value) MessageHeader

  { /* Send NOTIFY acc. to 34.229-1 Annex A.1.6 */

    var template (value) MessageHeader v_MessageHeader_Notify := cs_MessageHeader_Dummy;

    var charstring v_ToTag := f_SIP_SemicolonParamList_GetParamValue(p_SubscribeRequest.msgHeader.fromField.fromParams, c_tagId);

    var charstring v_NW_AddrStr := f_IMS_PTC_NW_Address_GetHostFormat();

    var charstring v_Branch1 := f_IMS_PTC_ImsInfo_GetNextBranch();

    var charstring v_Branch2 := f_IMS_PTC_ImsInfo_GetNextBranch();

    var charstring v_Scscf := px_IMS_Scscf;

    var integer v_Port_us := f_IMS_PTC_GetPort_us(not p_CondA2); 
    var template (value) ViaBody_List v_ViaBodyList;

    var template (value) SipUrl v_PublicUserId_TX := f_IMS_PTC_ImsInfo_GetPublicUserId_SipUrl_TX();

    var template (omit) GenericParam v_SigCompParam := f_IMS_PTC_ImsInfo_GetSigCompParam();

    var charstring v_ProtocolString := f_IMS_ConvertTransportProtocolToString(f_IMS_PTC_ImsInfo_GetTransportProtocol());

    v_ViaBodyList := {

      cs_ViaBody(v_ProtocolString, cs_HostPort(v_NW_AddrStr, v_Port_us), f_ViaParamsTX(v_Branch1, v_SigCompParam)),

      cs_ViaBody(v_ProtocolString, cs_HostPort(v_Scscf),                 f_ViaParamsTX(v_Branch2))

    };

    v_MessageHeader_Notify.via         := cs_Via(v_ViaBodyList);

    v_MessageHeader_Notify.fromField   := cs_From(v_PublicUserId_TX, tsc_IMS_ToTagSubscribeDialog);

    v_MessageHeader_Notify.toField     := cs_ToWithTag(v_PublicUserId_TX, v_ToTag);

    v_MessageHeader_Notify.callId      := p_SubscribeRequest.msgHeader.callId;

    v_MessageHeader_Notify.cSeq        := cs_CseqDef(p_CseqValue, "NOTIFY");

    v_MessageHeader_Notify.contact     := cs_Contact(cs_SipUrl_HostPort(v_Scscf));     // @sic R5s120727 change 10 sic@

    v_MessageHeader_Notify.contentType := cs_ContentTypeDef ("application/reginfo+xml");

    v_MessageHeader_Notify.event       := cs_EventDef("reg");

    v_MessageHeader_Notify.maxForwards := cs_MaxForwardsDef(69);

    v_MessageHeader_Notify.subscriptionState := p_SubscriptionState;

    return v_MessageHeader_Notify;

  }
…
After change

…

  function f_IMS_NotifyRequest_MessageHeader_A16(SUBSCRIBE_Request p_SubscribeRequest,

                                                 boolean p_CondA2,

                                                 integer p_CseqValue,

                                                 template (value) SubscriptionState p_SubscriptionState) runs on IMS_PTC return template (value) MessageHeader

  { /* Send NOTIFY acc. to 34.229-1 Annex A.1.6 */

    var template (value) MessageHeader v_MessageHeader_Notify := cs_MessageHeader_Dummy;

    var charstring v_ToTag := f_SIP_SemicolonParamList_GetParamValue(p_SubscribeRequest.msgHeader.fromField.fromParams, c_tagId);

    var charstring v_NW_AddrStr := f_IMS_PTC_NW_Address_GetHostFormat();

    var charstring v_Branch1 := f_IMS_PTC_ImsInfo_GetNextBranch();

    var charstring v_Branch2 := f_IMS_PTC_ImsInfo_GetNextBranch();

    var charstring v_Scscf := px_IMS_Scscf;

    var integer v_Port_ps := f_IMS_PTC_GetPort_ps(not p_CondA2); 
    var template (value) ViaBody_List v_ViaBodyList;

    var template (value) SipUrl v_PublicUserId_TX := f_IMS_PTC_ImsInfo_GetPublicUserId_SipUrl_TX();

    var template (omit) GenericParam v_SigCompParam := f_IMS_PTC_ImsInfo_GetSigCompParam();

    var charstring v_ProtocolString := f_IMS_ConvertTransportProtocolToString(f_IMS_PTC_ImsInfo_GetTransportProtocol());

    v_ViaBodyList := {

      cs_ViaBody(v_ProtocolString, cs_HostPort(v_NW_AddrStr, v_Port_ps), f_ViaParamsTX(v_Branch1, v_SigCompParam)),

      cs_ViaBody(v_ProtocolString, cs_HostPort(v_Scscf),                 f_ViaParamsTX(v_Branch2))

    };

    v_MessageHeader_Notify.via         := cs_Via(v_ViaBodyList);

    v_MessageHeader_Notify.fromField   := cs_From(v_PublicUserId_TX, tsc_IMS_ToTagSubscribeDialog);

    v_MessageHeader_Notify.toField     := cs_ToWithTag(v_PublicUserId_TX, v_ToTag);

    v_MessageHeader_Notify.callId      := p_SubscribeRequest.msgHeader.callId;

    v_MessageHeader_Notify.cSeq        := cs_CseqDef(p_CseqValue, "NOTIFY");

    v_MessageHeader_Notify.contact     := cs_Contact(cs_SipUrl_HostPort(v_Scscf));     // @sic R5s120727 change 10 sic@

    v_MessageHeader_Notify.contentType := cs_ContentTypeDef ("application/reginfo+xml");

    v_MessageHeader_Notify.event       := cs_EventDef("reg");

    v_MessageHeader_Notify.maxForwards := cs_MaxForwardsDef(69);

    v_MessageHeader_Notify.subscriptionState := p_SubscriptionState;

    return v_MessageHeader_Notify;

  }
…

